

# 第四章

## Black-Scholes 定價公式

第一節 歐式股票選擇權定價公式

第二節 BS 程式與繪圖

第三節 發放股利的調整

第四節 一般化的選擇權定價公式

本章中，我們將利用之前所學到的程式技巧，來撰寫 Black-Scholes 選擇權定價公式。包括陽春型的選擇權公式、發放離散股利的選擇權公式，以及發放連續股利型式的一般化的選擇權公式。

## 第一節 歐式股票選擇權定價公式

由於股票選擇權的價格與標的股票價格有直接的關連，因此在討論選擇權的訂價模型時，必須先對股票價格未來變動的走向做出適當的假設。在 BS 模型中，他們對股票價格的假設是，它遵行著所謂的擴散程序(Diffusion Process)。可以下式表示之

$$\frac{dS}{S} = \mu \cdot dt + \sigma \cdot dZ \dots\dots\dots(4.1.1)$$

其中

$$\frac{dS}{S} = \frac{S_{t+1} - S_t}{S_t} = \text{股票的報酬率。}$$

$\mu$  = 單位時間內預期的股價報酬率，為一常數。

$\sigma$  = 單位時間內預期的股價波動性，為一常數。

$dt$  = 單位時間。

$Z$  = 一隨機變數，為平均數為零，變異數為  $t$  之常態分配， $Z \sim N(0, t)$ 。

$Z$  稱之為韋恩程序(Wiener process)。

$dZ$  = 單位時間內， $Z$  的變動量，為一期望值為零，變異數為  $dt$  之常態分配， $dZ \sim N(0, dt)$ 。

(4.1.1)式的擴散程序又稱之為幾何布朗運動(Geometric Brownian Motion, GBM)。

由於衍生性金融商品的價格受到標的資產價格所影響，因此，它的風險主要來源即為標的資產。以股票選擇權為例，在 BSM 的論文中，便把買權價格  $C$  表示為是標的資產價格  $S$  與時間  $t$  的函數

$$C = f(S, t) \dots\dots\dots(4.1.2)$$

因此，買權價格  $C$  可視為因變數，標的資產價格  $S$  與時間  $t$  可視為自變數。

我們可將(4.1.2)式的買權價格，對標的股票價格取微分，令其為 Delta( $\Delta$ )

$$\Delta = \frac{\partial C}{\partial S} \dots\dots\dots(4.1.3)$$

Delta 代表相對於標的股票價格變動量下的買權價格變動量。因此，在 Delta 為 0.6 的情況下，如果股票價格變動 0.1 元，買權價格變動約為

$$dC = \Delta \times dS = 0.6 \times 0.1 = 0.06$$

由於(4.1.3)式是一個微分值，因此只有當股票價格的變動量很小的時候，才會成立，當股票價格的變動量很大的時候，便會產生誤差。

(4.1.3)式給我們一個很大的啟發，那就是我們可藉由選擇權部位和標的股票部位來建立一無風險的資產組合。以前例而言，如果我們同時賣出一單位買權、買入 0.6 單位的標的股票形成一個投資組合  $\Pi$ ，

$$\Pi = -C + 0.6S$$

則此投資組合在股票價格微小變動下是沒有風險的，因為  $\Delta = 0.6$ ，所以

$$d\Pi = -dC + 0.6 \times dS = -\Delta \times dS + 0.6 \times dS = 0。$$

這表示投資人可藉由選擇權部位和標的股票部位來建立一無風險資產組合。由於不存在套利的機會，此組合所產生的收益率應等同於無風險資產所能得到的報酬率，對一銀行而言，此即銀行的融資利率，也就是第二章第四節所說的以銀行信用為擔保的即期利率。

之所以能夠建立成一個無風險組合的原因，乃是由於股票價格和選擇權價格的不確定性有相同的來源：股價的變動。在短時間內，買權的價格和標的股票價格完全正相關，而賣權價格則和標的股票價格完全負相關。因此，適當的將股票和選擇權做一組合，則由股票產生的損益可完全抵消選擇權所產生的損益。所以在期末該組合的價值可以完全確定。

在 Black-Scholes 股票選擇權的公式中，他們基於無套利的條件，使用了風險中立的評價與(4.1.1)的股價隨機程序。其步驟為

1. 假設股價的預期收益為無風險資產的收益率  $r$ （亦即，假設  $\mu = r$ ）。
2. 計算選擇權在到期日時預期的償付。
3. 以無風險資產的收益率折現該償付。

藉由上述的分析，Black 和 Scholes 利用偏微分方程式，推導出了不支付股利股票的歐式買賣權公式。買權和賣權價格公式分別為

$$C = SN(d_1) - Ke^{-rT}N(d_2) \dots\dots\dots(4.1.4)$$

$$P = Ke^{-rT}N(-d_2) - SN(-d_1) \dots\dots\dots(4.1.5)$$

其中

$$d_1 = \frac{\ln(S/K) + (r + \sigma^2/2)T}{\sigma\sqrt{T}} \dots\dots\dots(4.1.6)$$

$$d_2 = \frac{\ln(S/K) + (r - \sigma^2/2)T}{\sigma\sqrt{T}} = d_1 - \sigma\sqrt{T} \dots\dots\dots(4.1.7)$$

$S$  = 目前股價，

$K$  = 執行價格，

$r$  = 融資利率為即期利率，

$T$  = 距到期日的時間，

$\sigma$  = 股價之波動性，

函數  $N(x)$  為標準常態變數之累積機率函數。

## 第二節 BS 程式與繪圖

我們將之前撰寫的標準常態分配機率密度函數、累積機率密度函數等程式碼，放在 **Module1** 中，以備其他程序的呼叫。下面重新列出這些程式碼。

```
#01 Const Pi = 3.14159265358979
#02
#03 Public Function NorPdf(d As Double) As Double
#04     Dim ans As Double
#05
#06     ans = Exp(-d * d / 2) / Sqr(2 * Pi)
#07
#08     NorPdf = ans
#09 End Function
#10
#11 Public Function NorCdf(d As Double) As Double
#12     Dim ans As Double, g As Double
#13     Const a1 = 0.4361836
#14     Const a2 = -0.1201676
#15     Const a3 = 0.937298
#16     g = 1 / (1 + 0.33267 * d)
#17
#18     If d >= 0 Then
#19         ans = 1 - (a1*g + a2*g*g + a3 *g*g*g) * NorPdf(d)
```

```

#20     Else
#21         ans = 1 - NorCdf(-d)
#22     End If
#23
#24     NorCdf = ans
#25 End Function
#26
#27 Public Function Op_Cvalue(S As Double, K As Double, _
#28     T As Double, sig As Double, r As Double) As Double
#29
#30     Dim d1 As Double, d2 As Double
#31
#32     d1 = (Log(S/K) + (r + sig*sig/2)* T) / (sig*Sqr(T))
#33     d2 = d1 - sig * Sqr(T)
#34
#35     Op_Cvalue = S * NorCdf(d1) - K * Exp(-r*T) * NorCdf(d2)
#36 End Function
#37
#38 Public Function Op_Pvalue(S As Double, K As Double, _
#39     T As Double, sig As Double, r As Double) As Double
#40
#41     Dim d1 As Double, d2 As Double
#42
#43     d1 = (Log(S/K) + (r + sig*sig/2)*T) / (sig*Sqr(T))
#44     d2 = d1 - sig * Sqr(T)
#45
#46     Op_Pvalue = K*Exp(-r*T)*NorCdf(-d2) - S * NorCdf(-d1)
#47 End Function

```

程式 4.1.1(Book1.xls/Module1)

#27~#36 為根據(4.1.4)式所撰寫的 BS 買權公式。在程式中我們分別計算出  $d_1$  與  $d_2$ ，#35 呼叫#11~#25 的 NorCdf 函數便完成計算。#38~#47 則為根據(4.1.5)式所撰寫的 BS 賣權公式。

為測試此程式，我們在工作表中放置(4.1.4)所需的資料如下圖，在 Sheet1 程式碼區撰寫下面的程式碼，由工作表中讀取資料，並呼叫 Module1 中的函數，將結果寫回工作表。並放置一個 Button 如上圖，將 Button 與下面 Op\_Calculate 程序連結。按 Button 執行下面程序，計算出選擇權的權利金。




圖 4.2.1(Book1.xls/Sheet1)

```

#01 Public Sub OP_Calculate()
#02     Dim S As Double
#03     Dim K As Double
#04     Dim T As Double
#05     Dim sig As Double
#06     Dim r As Double
#07     Dim y As Double
#08     Dim C_Value As Double
#09     Dim P_Value As Double
#10
#11     S = Worksheets("Sheet1").Cells(2, 1).Value
#12     K = Worksheets("Sheet1").Cells(2, 2).Value
#13     T = Worksheets("Sheet1").Cells(2, 3).Value
#14     sig = Worksheets("Sheet1").Cells(2, 4).Value
#15     r = Worksheets("Sheet1").Cells(2, 5).Value
#16
#17     C_Value = Op_Cvalue(S, K, T, sig, r)
#18     P_Value = Op_Pvalue(S, K, T, sig, r)
#19
#20     Worksheets("Sheet1").Cells(5, 2).Value = C_Value
#21     Worksheets("Sheet1").Cells(5, 3).Value = P_Value
#22 End Sub

```

程式 4.1.2(Book1.xls/Sheet1)

## 一、合併 Put 與 Call 的計算

程式 4.1.1 中，我們把 Call 與 Put 權利金的計算函數，分別撰寫。事實上，我們可以合併為一個函數。下面的 BSOOption 便透過傳遞的 OpClass 參數，藉由控制變數  $z$ ，來決定計算的是 Put 或 Call 的權利金。當計算的對象是 Call， $z = 1$ ，計算的對象是 Put， $z = -1$ 。這是由於(4.1.4)與(4.1.5)，有對稱的關係存在，所以才可以用如此處理。

```
#01 Public Function BSOOption(OpClass As String, _  
#02     S As Double, K As Double, T As Double, _  
#03     sig As Double, r As Double) As Double  
#04  
#05     Dim d1 As Double, d2 As Double  
#06     Dim z As Double  
#07  
#08     If OpClass = "C" Then  
#09         z = 1#  
#10     Else  
#11         z = -1#  
#12     End If  
#13  
#14     d1 = (Log(S/K) + (r + sig*sig/2)*T) / (sig*Sqr(T))  
#15     d2 = d1 - sig * Sqr(T)  
#16  
#17     BSOOption = z * (S * NorCdf(z*d1) - K*Exp(-r*T) * _  
#18         NorCdf(z*d2))  
#19  
#20 End Function
```

程式 4.1.3(Book1.xls/Module1)



程式 4.1.4 中，為呼叫程式 4.1.3 的範例。#16 與#17 分別為計算 Call 與 Put 的呼叫。第一個參數分別傳入 “C” 或 “P”，來決定計算的是 Put 或 Call 的權利金。

```
#01 Public Sub Op_Calculate_1()  
#02     Dim S As Double  
#03     Dim K As Double  
#04     Dim T As Double  
#05     Dim sig As Double  
#06     Dim r As Double  
#07     Dim C_Value As Double  
#08     Dim P_Value As Double  
#09  
#10     S = Worksheets("Sheet2").Cells(2, 1).Value  
#11     K = Worksheets("Sheet2").Cells(2, 2).Value  
#12     T = Worksheets("Sheet2").Cells(2, 3).Value  
#13     sig = Worksheets("Sheet2").Cells(2, 4).Value  
#14     r = Worksheets("Sheet2").Cells(2, 5).Value  
#15  
#16     C_Value = BSOOption("C", S, K, T, sig, r)  
#17     P_Value = BSOOption("P", S, K, T, sig, r)  
#18  
#19     Worksheets("Sheet2").Cells(5, 2).Value = C_Value  
#20     Worksheets("Sheet2").Cells(5, 3).Value = P_Value  
#21 End Sub
```

程式 4.1.4(Book1.xls/sheet2)

下圖為配合 For...Next 迴圈，連續計算在不同的資產價格之下，買權與賣權的權利金，並以 Excel 的繪圖功能，畫出 Call 與 Put 的價格線圖。




圖 4.2.2(Book2.xls/Sheet1)

資產價格分別放在 C2 到 C32 的儲存格內，因此計算迴圈要依序讀取價格資料。程式 4.1.5 為此計算程式，#16 到#22 為此計算迴圈，利用 Cells 物件的列參數的變動，Cells(i, 3)，由 i = 2 到 32，完成資料的讀取。並在迴圈中寫回試算表。

```
#01 Public Sub OP_Calculate()  
#02     Dim i As Integer  
#03     Dim S As Double  
#04     Dim K As Double  
#05     Dim T As Double  
#06     Dim sig As Double  
#07     Dim r As Double  
#08     Dim C_Value As Double  
#09     Dim P_Value As Double  
#10  
#11     K = Worksheets("Sheet1").Cells(1, 2).Value  
#12     T = Worksheets("Sheet1").Cells(2, 2).Value  
#13     sig = Worksheets("Sheet1").Cells(3, 2).Value  
#14     r = Worksheets("Sheet1").Cells(4, 2).Value  
#15  
#16     For i = 2 To 32  
#17         S = Worksheets("Sheet1").Cells(i, 3).Value  
#18         C_Value = BSOOption("C", S, K, T, sig, r)  
#19         P_Value = BSOOption("P", S, K, T, sig, r)  
#20         Worksheets("Sheet1").Cells(i,5).Value=C_Value  
#21         Worksheets("Sheet1").Cells(i,6).Value=P_Value  
#22     Next i  
#23 End Sub
```

程式 4.1.5(Book2.xls/Sheet1)

下圖為買權與賣權的圖形，利用 Excel 的計算與繪圖功能，我們可以輕易的製作出權利金圖，以及相關 Greeks 圖形，這將於下面的章節中說明。




圖 4.2.3 買權權利金圖




圖 4.2.4 賣權權利金圖



### 第三節 發放股利的調整

第一節的選擇權公式只適用於沒有發放股利的情況，針對股利的發放需作必要的調整，分別以間段與連續股利來考慮。

#### 一、間斷股利形式

如果在到期日前的 $t$ 時， $t < T$ ，股票發放一固定的股利 $D_t$ ，則我們以無風險資產收益率將之折現(為什麼?)，並由目前股價中扣除此股利的現值，

$D_t e^{-rt}$ 。令

$$S_D = S - D_t e^{-rt} \dots\dots\dots(4.3.1)$$

其中 $S_D$ 為股利調整過的目前資產價格。將之代入BS模式中，取代公式中的股價 $S$ ，我們便可求得考量股利後的選擇權價格。

由於此股票在到期前支付股利，因此BS公式可修正為

$$C = S_D N(d_1) - K e^{-rT} N(d_2) \dots\dots\dots(4.3.2)$$

$$d_1 = \frac{\ln(S_D/K) + (r + \sigma^2/2)T}{\sigma\sqrt{T}}$$

$$d_2 = \frac{\ln(S_D/K) + (r - \sigma^2/2)T}{\sigma\sqrt{T}} = d_1 - \sigma\sqrt{T}$$

程式 4.1.6 為間斷股利選擇權計算的範例，我們並沒有另寫函數去計算權利金。作法是在呼叫原來的 BSOOption 前，先依照(4.3.1)調整股價。如此，便可得到正確的結果。

```
#01 Public Sub Op_Calculate_2()
#02     Dim S As Double
#03     Dim Sd As Double
#04     Dim K As Double
#05     Dim T As Double
#06     Dim sig As Double
#07     Dim r As Double
#08     Dim D As Double
#09     Dim td As Double
#10     Dim C_Value As Double
#11     Dim P_Value As Double
#12
#13     S = Worksheets("Sheet3").Cells(2, 1).Value
#14     K = Worksheets("Sheet3").Cells(2, 2).Value
#15     T = Worksheets("Sheet3").Cells(2, 3).Value
#16     sig = Worksheets("Sheet3").Cells(2, 4).Value
#17     r = Worksheets("Sheet3").Cells(2, 5).Value
#18     D = Worksheets("Sheet3").Cells(2, 6).Value
#19     td = Worksheets("Sheet3").Cells(2, 7).Value
#20
#21     Sd = S - Exp(-r * td)
#22
#23     C_Value = BSOOption("C", Sd, K, T, sig, r)
#24     P_Value = BSOOption("P", Sd, K, T, sig, r)
#25
#26     Worksheets("Sheet3").Cells(5,2).Value=C_Value
#27     Worksheets("Sheet3").Cells(5,3).Value=P_Value
#28 End Sub
```

程式 4.1.6(Book1.xls/Sheet3)

#21 便是依照(4.3.1)式調整股價，再將之代入(4.1.4)與(4.1.5)式中，標的資產價格的位置即可。

## 二、連續股利形式

對於一個發放連續固定股利比率  $y$  的股票，我們可如下調整  $S$ ，

$$S_y = Se^{-yT} \quad (4.3.3)$$

將上式的  $S_y$  代入 BS 公式中，取代所有的股價  $S$ 。如此便可求出連續股利的選擇權價格了。

```
#01 Public Sub Op_Calculate_3()  
#02     Dim S As Double  
#03     Dim Sd As Double  
#04     Dim K As Double  
#05     Dim T As Double  
#06     Dim sig As Double  
#07     Dim r As Double  
#08     Dim y As Double  
#09     Dim C_Value As Double  
#10     Dim P_Value As Double  
#11  
#12     S = Worksheets("Sheet4").Cells(2, 1).Value  
#13     K = Worksheets("Sheet4").Cells(2, 2).Value  
#14     T = Worksheets("Sheet4").Cells(2, 3).Value  
#15     sig = Worksheets("Sheet4").Cells(2, 4).Value  
#16     r = Worksheets("Sheet4").Cells(2, 5).Value  
#17     y = Worksheets("Sheet4").Cells(2, 6).Value  
#18  
#19     Sd = S * Exp(-y * T)  
#20  
#21     C_Value = BSOOption("C", Sd, K, T, sig, r)  
#22     P_Value = BSOOption("P", Sd, K, T, sig, r)  
#23  
#24     Worksheets("Sheet4").Cells(5, 2).Value = C_Value  
#25     Worksheets("Sheet4").Cells(5, 3).Value = P_Value  
#26 End Sub
```

程式 4.1.7(Book1.xls/Sheet4)

#19 便是依照(4.3.3)式調整股價，在將之代入(4.1.4)與(4.1.5)式中，標的資產價格的位置即可。



## 第四節 一般化的選擇權定價公式

不論是指數選擇權、外幣選擇權、債券選擇權，持有這些選擇權的標的資產都可取得一定的收益，雖然這些標的資產的實質內容不同，但就選擇權契約的形式而言，確是沒有任何的差異。(4.3.3)的連續固定股利比率  $y$ ，可以解釋為標的資產的收益率，如此便可適用於外幣選擇權和債券選擇權。

將(4.3.1)的調整，代入(4.1.4)與(4.1.5)式，得到修正後的 BS 公式，分別成為

$$C = Se^{-yT} N(d_1) - Ke^{-rT} N(d_2) \dots\dots\dots(4.4.1)$$

$$P = Ke^{-rT} N(-d_2) - Se^{-yT} N(-d_1) \dots\dots\dots(4.4.2)$$

$$d_1 = \frac{\ln(S/K) + (r + \sigma^2/2)T}{\sigma\sqrt{T}} = \frac{\ln(Se^{-yT}/K) + (r + \sigma^2/2)T}{\sigma\sqrt{T}}$$

$$= \frac{\ln(S/K) + (r - y + \sigma^2/2)T}{\sigma\sqrt{T}}$$


$$d_2 = \frac{\ln(S/K) + (r - y - \sigma^2/2)T}{\sigma\sqrt{T}} = d_1 - \sigma\sqrt{T}$$

程式 4.1.8 為(4.4.1)與(4.4.2)一般化選擇權權利金的計算函數。當我們令  $y = 0$ ，便會回到不發放股利的 BSOOption 計算函數。因此，往後我們的計算呼叫，將以 GBSOption 取代 BSOOption。

```
#01 Public Function GBSOption(OpClass As String, _  
#02     S As Double, K As Double, T As Double, sig As _  
#03     Double, r As Double, y As Double) As Double  
#04  
#05     Dim d1 As Double, d2 As Double  
#06     Dim z As Double  
#07  
#08     If OpClass = "C" Then  
#09         z = 1#  
#10     Else  
#11         z = -1#  
#12     End If  
#13  
#14     d1 = (Log(S/K) + (r-y+sig*sig/2)*T) / (sig*Sqr(T))  
#15     d2 = d1 - sig * Sqr(T)  
#16  
#17     GBSOption = z * (S * Exp(-y*T) * NorCdf(z*d1) _  
#18         - K * Exp(-r*T) * NorCdf(z*d2))  
#19  
#20 End Function
```

程式 4.1.8(Sheet1.xls/Module1)

## 財務工程與EXCEL VBA 的應用 - 選擇權評價理論之實作 ( 附範例光碟 )

作者：董夢雲 [追蹤作者](#) 

出版社：證期會 [訂閱出版社新書快訊](#) 

出版日期：2005/08/01

語言：繁體中文

定價：560元

優惠價：95折 532元

本商品單次購買10本9折 504元



金融研訓院 特約講師  
證券暨投資分析人員合格(CSIA)  
希奇資本 技術長(CTO)

董 夢 雲 財務博士

Mobil: (Taiwan)0988-065-751 (China)1508-919-2872

EEmail: dongmy@ms5.hinet.net

Line ID/WeChat ID: andydong3137

#### 專長

GPU 平行運算與財務工程，C#、.Net Framework、CUDA、OpenCL、C、C++。

外匯與利率結構商品評價實務，股權與債權及衍生品評價實務。

風險管理理論與實務，資本配置與額度規劃。

#### 經歷

中國信託商業銀行交易室研發科主管

凱基證券風險管理部主管兼亞洲區風險管理主管

中華開發金控、工業銀行風險管理處處長

永豐金控、商業銀行風險管理處處長

永豐商業銀行結構商品開發部副總經理

#### 著作

金融選擇權：市場、評價與策略，第二版，1997，新陸書局。

財務工程與 Excel VBA 的應用：選擇權評價理論之實作，2005，證券暨期貨發展基金會。

#### 翻譯

衍生性金融商品與內部稽核，2003，金融研訓院。